Что-то в этом мире не так...
Есть КА 2.4...
Есть 2 документа "План производства" и "Производство без заказа"...
Для простоты начального этапа созданы 2 документа
"План производства" с тремя номенклатурами и "Производство без заказа" с двумя...

```1c
    ВЫБРАТЬ
    ПланПроизводства_Продукция.Ссылка КАК Ссылка,
    ПланПроизводства_Продукция.Номенклатура КАК Номенклатура,
    ПланПроизводства_Продукция.Количество КАК Количество,
    0 КАК КоличествоПро
    ИЗ
    Документ.ПланПроизводства.Продукция КАК ПланПроизводства_Продукция
    ГДЕ
    ПланПроизводства_Продукция.Ссылка = &Ссылка
    И ПланПроизводства_Продукция.Количество <> 0

    ОБЪЕДИНИТЬ

    ВЫБРАТЬ
    ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка,
    ПроизводствоБезЗаказа_ВыходныеИзделия.Номенклатура,
    0,
    ПроизводствоБезЗаказа_ВыходныеИзделия.Количество
    ИЗ
    Документ.ПроизводствоБезЗаказа.ВыходныеИзделия КАК ПроизводствоБезЗаказа_ВыходныеИзделия
    ГДЕ
    ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез
```

этот запрос выдает
Ссылка Номенклатура Количество КоличествоПро
План производства ДН-00000064 от 01.08.2019 0:00:00 СКОВОРОДА 22 ПР ЛИТАЯ "Надежная" (полимер-кер.) с/ручка 100
План производства ДН-00000064 от 01.08.2019 0:00:00 Сковорода-гриль 26 ПР ЛИТАЯ "Надежная" (полимер-керам.)со с/р 90
План производства ДН-00000064 от 01.08.2019 0:00:00 СКОВОРОДА 28 ПР ЛИТАЯ "Бордо" (полимер-керам.) 80
Производство без заказа 00ДН-000271 от 09.10.2019 17:35:52 СКОВОРОДА 22 ПР ЛИТАЯ "Надежная" (полимер-кер.) с/ручка 20
Производство без заказа 00ДН-000271 от 09.10.2019 17:35:52 Сковорода-гриль 26 ПР ЛИТАЯ "Надежная" (полимер-керам.)со с/р 20

А тот что ниже не показывает третью строку с количеством 80!!!

```1c
    ВЫБРАТЬ
    ПланПроизводства_Продукция.Номенклатура КАК Номенклатура,
    ПланПроизводства_Продукция.Количество КАК Количество,
    ЕСТЬNULL(ПроизводствоБезЗаказа_ВыходныеИзделия.Количество, 0) КАК КоличествоПро
    ИЗ
    Документ.ПланПроизводства.Продукция КАК ПланПроизводства_Продукция
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.ПроизводствоБезЗаказа.ВыходныеИзделия КАК ПроизводствоБезЗаказа_ВыходныеИзделия
    ПО ПланПроизводства_Продукция.Номенклатура = ПроизводствоБезЗаказа_ВыходныеИзделия.Номенклатура
    ГДЕ
    ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез
    И ПланПроизводства_Продукция.Ссылка = &Ссылка
    И ПланПроизводства_Продукция.Количество <> 0
```

Номенклатура Количество КоличествоПро
СКОВОРОДА 22 ПР ЛИТАЯ "Надежная" (полимер-кер.) с/ручка 100,000 20
Сковорода-гриль 26 ПР ЛИТАЯ "Надежная" (полимер-керам.)со с/р 90,000 20

ПОЧЕМУ НЕ ВЫВОДИТСЯ ТРЕТЬЯ СТРОКА ПРИ ЛЕВОМ СОЕДИНЕНИИ?
---
Вы отсекаете эту строку условием ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез

Попробуйте поставить его в само соединение примерно вот так:

```1c
    ВЫБРАТЬ
    ПланПроизводства_Продукция.Номенклатура КАК Номенклатура,
    ПланПроизводства_Продукция.Количество КАК Количество,
    ЕСТЬNULL(ПроизводствоБезЗаказа_ВыходныеИзделия.Количество, 0) КАК КоличествоПро
    ИЗ
    Документ.ПланПроизводства.Продукция КАК ПланПроизводства_Продукция
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.ПроизводствоБезЗаказа.ВыходныеИзделия КАК ПроизводствоБезЗаказа_ВыходныеИзделия
    ПО ПланПроизводства_Продукция.Номенклатура = ПроизводствоБезЗаказа_ВыходныеИзделия.Номенклатура
    И (ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез)
    ГДЕ
    И ПланПроизводства_Продукция.Ссылка = &Ссылка
    И ПланПроизводства_Продукция.Количество <> 0
```
---
В этом мире все ТАК как вы настроите...
Вы настроли неверно и получили неверный результат! Всего лишь... Мир ни при чем... ;-)))
---
Вы отсекаете эту строку условием ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез

Попробуйте поставить его в само соединение примерно вот так:

```1c
    ВЫБРАТЬ
    ПланПроизводства_Продукция.Номенклатура КАК Номенклатура,
    ПланПроизводства_Продукция.Количество КАК Количество,
    ЕСТЬNULL(ПроизводствоБезЗаказа_ВыходныеИзделия.Количество, 0) КАК КоличествоПро
    ИЗ
    Документ.ПланПроизводства.Продукция КАК ПланПроизводства_Продукция
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.ПроизводствоБезЗаказа.ВыходныеИзделия КАК ПроизводствоБезЗаказа_ВыходныеИзделия
    ПО ПланПроизводства_Продукция.Номенклатура = ПроизводствоБезЗаказа_ВыходныеИзделия.Номенклатура
    И (ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез)
    ГДЕ
    И ПланПроизводства_Продукция.Ссылка = &Ссылка
    И ПланПроизводства_Продукция.Количество <> 0
```
---
Условие, налагаемое на присоединяемую таблицу в секции ГДЕ, превращает левое соединение во внутреннее.
---
Спасибо!
Boyborodin за решение....
antz за дополнительную консультацию...
YannikAlx за моральную поддердку....

А так как Boyborodin сделел маленькую ошибку - не в консоли писано скорее всего...
В секцмм ГДЕ лишняя "И"...

Привожу запрос

```1c
    ВЫБРАТЬ
    ПланПроизводства_Продукция.Номенклатура КАК Номенклатура,
    ПланПроизводства_Продукция.Количество КАК Количество,
    ЕСТЬNULL(ПроизводствоБезЗаказа_ВыходныеИзделия.Количество, 0) КАК КоличествоПро
    ИЗ
    Документ.ПланПроизводства.Продукция КАК ПланПроизводства_Продукция
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.ПроизводствоБезЗаказа.ВыходныеИзделия КАК ПроизводствоБезЗаказа_ВыходныеИзделия
    ПО ПланПроизводства_Продукция.Номенклатура = ПроизводствоБезЗаказа_ВыходныеИзделия.Номенклатура
    И (ПроизводствоБезЗаказа_ВыходныеИзделия.Ссылка = &СсылкаПроБез)
    ГДЕ
    ПланПроизводства_Продукция.Ссылка = &Ссылка
    И ПланПроизводства_Продукция.Количество <> 0
```
---
Да, быстренько накидал ручками, по этому и написал что решение "примерно" )) Главное, что вам помогло.